A Further Spreadsheet-Based Illustration of the Convergence of the Cox-Ross-Rubinstein Binomial Option Pricing Model to the Black-Scholes-Merton Version

Clarence C.Y. Kwan
DeGroote School of Business
McMaster University
Hamilton, Ontario L8S 4M4
Canada
E-mail: kwanc@mcmaster.ca

August 2025

Abstract

The Cox-Ross-Rubinstein binomial option pricing model, which retains the economic insights of the Black-Scholes-Merton version without requiring the use of advanced mathematical tools, is suitable for coverage in finance courses where the topic of options is covered. If the number of time steps in the Cox-Ross-Rubinstein model between the valuation date and the expiry date of an option approaches infinity, the two models will converge on statistical grounds. This pedagogic note addresses some computational issues as the number of time steps increases to high levels. It also provides a measure to counteract potential deteriorations of computational precision. The spreadsheet-based illustration in this note complements very well the statistical justification for the convergence of the two models.

Keywords: Black-Scholes-Merton option pricing model, Cox-Ross-Rubinstein binomial option pricing model, parameter matching for model convergence, computational precision

A Further Spreadsheet-Based Illustration of the Convergence of the Cox-Ross-Rubinstein Binomial Option Pricing Model to the Black-Scholes-Merton Version

1 Introduction

The Black and Scholes [1973] option pricing model is also known as the Black-Scholes-Merton option pricing model, in recognition of the contributions of Merton [1973]. The model (hereafter, BSM-OPM) has profound impacts on finance and investments. The BSM-OPM has inspired many innovations in academic research. Furthermore, accompanying improvements of valuation techniques for derivative assets in practice, there has been phenomenal growth in derivatives markets and in the types of derivatives-based investment products.

The Black-Scholes derivation of the BSM-OPM requires advanced mathematical tools to solve a partial differential equation. The alternative derivation by Merton [1973], known as risk-neutral valuation, is much simpler as it does not require any differential equations to be solved. The Merton approach, which is applicable to valuation of various derivative assets, has been replicated in Hull [2018], as part of an extensive coverage of option pricing. However, the analytical details of the Merton approach, which still require mathematical tools that are unfamiliar to many finance students, are seldom part of the general finance curriculum at undergraduate or graduate levels. Nevertheless, given the significance of the BSM-OPM in finance, the corresponding pricing formula for call options is routinely covered in standard finance textbooks.[1]

Cox, Ross, and Rubinstein [1979] (hereafter, CRR) have used instead a binomial approach to derive an option pricing model (hereafter, CRR-BOPM). Detailed coverage of the CRR-BOPM tends to be available in advanced textbooks only; however, brief descriptions of the approach are still available in some standard textbooks.[2] The key mathematical requirement for the approach is the binomial theorem, which is intuitive for students who are familiar with the growth pattern of Pascal's triangle. A common feature in the derivations of the BSM-OPM and

[1]See, for example, Berk and DeMarzo [2020, Chapter 21] and Ross, Westerfield, Jaffe, and Jordan [2022, Chapter 24] for U.S. editions, and Berk, DeMarzo, and Stangeland [2022, Chapter 15] and Ross, Westerfield, Jaffe, and Driss [2022, Chapter 24] for the corresponding Canadian editions.

[2]See, for example, Copeland, Weston, and Shastri [2005, Chapter 7] and Hull [2018, Chapter 13] for detailed coverage of the model, and the references in footnote 1 for some brief descriptions.

the CRR-BOPM — which are continuous-time and discrete-time models, respectively — is that both use the same economic intuition of risk-free hedges during the derivation process. The key difference between the two derived results is in the probability distributions involved; specifically, it is the standard normal distribution for the BSM version, but the binomial distribution instead for the CRR version.

Students learn in statistics that, for a given probability of success in repeated binomial trials, as the number of trials increases, the closer is the binomial distribution to a normal distribution, and that the convergence of the two distributions requires infinitely many binomial trials. For the convergence to be meaningful in the context of option pricing, the underlying parameters in the two models must match properly. CRR have provided a specific relationship for the parameters in the two models as required for an exact match. Such a relationship, which has been justified on statistical grounds, ensures an exact match if there are infinitely many time steps between the valuation date and the expiry date of each option in the CRR-BOPM.

Among the 11 versions of binomial option pricing models reviewed in Chance [2008], the CRR-BOPM is, by far, the simplest model in terms of parameter matching and thus is the best choice for pedagogic illustrations of the convergence issue. Feng and Kwan [2012] have replicated the original derivation of the CRR-BOPM for call options and have illustrated its convergence to the BSM-OPM from a pedagogic perspective. The emphasis of the Microsoft Excel-based illustrations there is on the efficiency of convergence in terms of the required time steps to achieve an acceptable approximation. Specifically, convergence is considered to have been achieved if the difference between the corresponding option values based on the CRR-BOPM and the BSM-OPM is within \$0.010 of each other.

This pedagogic note provides a further Excel-based illustration of the convergence of the two models. As the prices of the corresponding European call and put options on stocks that pay no dividends are related by put-call parity, the illustration here also covers such put options.[3] Instead of relying on a specific convergence criterion, this note considers large numbers of time steps between the valuation date and the expiry date of each option given that the Excel function BINOM.DIST for binomial distributions can accommodate as many as $2,147,483,646$ ($= 2^{31} - 2$) binomial trials. The illustration here complements very well the statistical justification for the convergence of the two models, which requires the use of infinitely many time steps.

[3]See, for example, Copeland, Weston, and Shastri [2005, Chapter 7] for a derivation of put-call parity.

Excel can store numbers from approximately 2.2×10^{-308} to approximately 1.8×10^{308} with 15 digits of precision.[4] As explained in Section 2, if a large number of time steps is used to illustrate the convergence of the CRR-BOPM to the BSM-OPM, some computational issues need to be addressed. To illustrate, suppose that there are 10^9 time steps between the valuation date and the expiry date of an option. Given Excel's 15-digit precision, if the computed risk-free interest rate in each time step is $r_f = 1.23456789012345 \times 10^{-11}$, the corresponding value of $1 + r_f$, which is needed to compute the probability of success in a binomial trial for the Excel function BINOM.DIST, is stored as 1.00000000001234. The loss of the remaining digits will cause deteriorations in precision for subsequent computations.

Given Excel's 15-digit precision, Excel-based illustrations of the convergence of the CRR-BOPM to the BSM-OPM with time steps in the order of 100, as provided in Feng and Kwan [2012], have allowed students to visualize how increases in time steps affect the difference in the corresponding option prices. However, for time steps of higher dimensions, the issue of potential deteriorations of the computed results based on CRR-BOPM needs to be addressed. While still staying with CRR-BOPM, this note provides some remedial measures to counteract potential deteriorations of computational precision if the number of time steps is approaching or has reached the computational limit of the Excel function BINOM.DIST. The details are provided in Section 3. The Excel file accompanying this note, which allows users to attempt different sets of input parameters without concerns about problematic computational issues, is suitable for further illustrations of the convergence of the two models.

2 Black-Scholes-Merton and Cox-Ross-Rubinstein Pricing Formulas

This section first replicates the derived results of the BSM-OPM and the CRR-BOPM for European call options on a stock that pays no dividends. It then replicates the corresponding results for put options. Parameter matching to ensure the convergence of the two models is considered separately in Section 3. To facilitate the Excel-based illustration in Section 4, computational issues arising from the use of large numbers of time steps between the valuation date and the expiry date of an option in the CRR-BOPM are addressed in Section 3, along with

[4] See https://learn.microsoft.com/en-us/office/troubleshoot/excel/floating-point-arithmetic-i.

a measure to counteract potential deteriorations of computational precision.

2.1 The Black-Scholes-Merton Formula for Call Options

Let S be the stock price on the valuation date, σ be the stock's standard deviation of annual returns, r be the continuously compounded annual risk-free interest rate, and X be the exercise price (also known as the strike price) of a call option, which expires in proportion T of a year. According to the BSM-OPM, the call option price is

$$\mathbf{C} = S\,\mathbf{N}(d_1) - X\,e^{-rT}\,\mathbf{N}(d_2), \tag{1}$$

where

$$d_1 = \frac{1}{\sigma\sqrt{T}}\left[\ln\left(\frac{S}{X}\right) + rT\right] + \frac{1}{2}\sigma\sqrt{T} \tag{2}$$

and

$$d_2 = d_1 - \sigma\sqrt{T}. \tag{3}$$

In the standard normal distribution function $f(z)$, $\mathbf{N}(d_1)$, for example, is the the area enclosed by $f(z)$ and the z-axis from $z = -\infty$ to $z = d_1$; that is,

$$\mathbf{N}(d_1) = \int_{z=-\infty}^{d_1} f(z)dz.$$

2.2 The Cox-Ross-Rubinstein Formula for Call Options

In addition to S and X as defined earlier, let N be the number of time steps between the valuation date and the expiry date of the same European call option, u and d be the multiplicative factors for up-down price movements in each time step, and r_f be the risk-free interest rate in each time step. The last three parameters above are required to satisfy the condition of

$$0 < d < 1 + r_f < u. \tag{4}$$

The following tree diagrams for $N = 1$ and $N = 2$, which can be extended successively for greater numbers of time steps, provide a visual illustration of what underlies the CRR-BOPM:

$$\begin{array}{ccccccc} & & uS & & & & C_u = \max(0, uS - X) \\ & \nearrow & & & & \nearrow & \\ S & & & & C & & \\ & \searrow & & & & \searrow & \\ & & dS & & & & C_d = \max(0, dS - X) \end{array}$$

$$
\begin{array}{ccccc}
 & & & & uuS \\
 & & & / & \\
 & & uS & & \\
 & / & & \backslash & \\
S & & & & udS \\
 & \backslash & & / & = duS \\
 & & dS & & \\
 & & & \backslash & \\
 & & & & ddS
\end{array}
\qquad
\begin{array}{ccccl}
 & & & & C_{uu} \\
 & & & / & = \max(0, u^2S - X) \\
 & & C_u & & \\
 & / & & \backslash & \\
C & & & & C_{ud} = C_{du} \\
 & \backslash & & / & = \max(0, udS - X) \\
 & & C_s & & \\
 & & & \backslash & \\
 & & & & C_{dd} \\
 & & & & = \max(0, d^2S - X)
\end{array}
$$

As described in Feng and Kwan [2012], for example, a risk-free hedge based on the call option and the underlying stock is formed at the beginning of each time step. For $N = 1$, the risk-free hedge directly leads to

$$C = \frac{pC_u + (1-p)C_d}{1 + r_f} = \frac{p \max(0, uS - X) + (1-p)\max(0, dS - X)}{1 + r_f},$$

where

$$p = \frac{(1 + r_f) - d}{u - d} \tag{5}$$

and

$$1 - p = \frac{u - (1 + r_f)}{u - d}. \tag{6}$$

As long as inequality (4) holds, the conditions of $0 < p < 1$ and $0 < 1 - p < 1$ are assured.

For $N = 2$, the two successive risk-free hedges lead to

$$C = \frac{pC_u + (1-p)C_d}{1 + r_f}, \tag{7}$$

$$C_u = \frac{pC_{uu} + (1-p)C_{ud}}{1 + r_f} = \frac{p \max(0, u^2S - X) + (1-p)\max(0, udS - X)}{1 + r_f},$$

and

$$C_d = \frac{pC_{du} + (1-p)C_{dd}}{1 + r_f} = \frac{p \max(0, udS - X) + (1-p)\max(0, d^2S - X)}{1 + r_f}.$$

Note that, for computing p and $1 - p$ via equations (5) and (6), respectively, the corresponding u, d, and r_f are for each time step instead. After eliminating C_u and C_d from equation (7), we have, more directly,

$$C = \frac{p^2 \max(0, u^2S - X) + 2p(1-p)\max(0, udS - X) + (1-p)^2 \max(0, d^2S - X)}{(1 + r_f)^2}.$$

Extending the same idea for $N = 1$ and $N = 2$ to the case of a general number of time steps requires the use of the binomial theorem. The resulting expression is

$$C = \left[\sum_{n=0}^{N} \frac{N!}{(N-n)!n!} p^n (1-p)^{N-n} \max\left(0, u^n d^{N-n} S - X\right)\right] \frac{1}{(1+r_f)^N}, \tag{8}$$

where

$$n! = n(n-1)(n-2)\cdots(2)(1),$$

for $n \geq 1$ and $0! = 1$. However, to facilitate a direct comparison between the CRR-BOPM and the BSM-OPM, there is a need to rewrite equation (8) without the $\max\left(0, u^n d^{N-n} S - X\right)$ terms.

With integer a denoted as the smallest n for which

$$u^n d^{N-n} S - X > 0$$

or, equivalently,

$$n > \ln\left(\frac{X}{d^N S}\right) \Big/ \ln\left(\frac{u}{d}\right), \tag{9}$$

equation (8) becomes

$$C = S\ B(n \geq a|N, p') - \frac{X}{(1+r_f)^N}\ B(n \geq a|N, p), \tag{10}$$

where

$$p' = \frac{pu}{1+r_f}. \tag{11}$$

Just like parameter p as defined in equation (5), parameter p', which satisfies the condition of

$$0 < p < p' < 1,$$

also takes the role of a probability. The discrete functions

$$B(n \geq a|N, p) = \sum_{n=a}^{N} \frac{N!}{(N-n)!n!} p^n (1-p)^{N-n}$$

and

$$B(n \geq a|N, p') = \sum_{n=a}^{N} \frac{N!}{(N-n)!n!} (p')^n (1-p')^{N-n}$$

are complementary binomial distribution functions.

For the CRR-BOPM to work as intended, the integer a as deduced from inequality (9) is required to satisfy the condition of

$$0 \leq a \leq N. \tag{12}$$

However, if the call option is deep in the money or deep out of the money, for which S and X are very far apart, inequality (12) may be violated if N is not large enough to avoid such occurrences. Violations of inequality (12) will lead to computational failures. This issue is addressed later in Section 3 where the convergence of the CRR-BOPM to the BSM-OPM and the corresponding computational issues are considered.

2.3 Black-Scholes-Merton and Cox-Ross-Rubinstein Formulas for Put Options

According to put-call parity, the difference between the corresponding call and put option prices is the same as the difference between the underlying stock price and the present value of the exercise price. The continuous-time version of put-call parity is

$$\mathbf{C} - \mathbf{P} = S - X\, e^{-rT}, \tag{13}$$

where $\mathbf{P}$ is the put option price. The BSM formula for the put option, therefore, is

$$\mathbf{P} = X\, e^{-rT}\, \mathbf{N}(-d_2) - S\, \mathbf{N}(-d_1). \tag{14}$$

The discrete-time version of put-call parity is

$$C - P = S - \frac{X}{(1 + r_f)^N} \tag{15}$$

instead, where P is the put option price. The corresponding CRR formula for the put option, therefore, is

$$P = \frac{X}{(1 + r_f)^N}\, B(n < a|N, p) - S\, B(n < a|N, p'). \tag{16}$$

Here, $B(n < a|N, p)$ and $B(n < a|N, p')$ are cumulative binomial distributions. The same issue pertaining to the computed integer a for deep-in-the-money and deep-out-of-the money call options, which can cause violations of inequality (12), as raised at the end of the preceding subsection, is also applicable to the corresponding put option.

3 Parameter Matching and Computational Issues

To match the continuously compounded risk-free interest rate r in the BSM-OPM, where the option expires in proportion T of a year, and the risk-free interest rate r_f over each of the N time steps in CRR-BOPM, requires[5]

$$(1+r_f)^N = \exp(rT) \tag{17}$$

or, equivalently,

$$1 + r_f = \exp\left(\frac{rT}{N}\right). \tag{18}$$

Thus, for call options, the convergence of C to $\mathbf{C}$ requires that both the difference between $B(n \geq a|N, p')$ and $\mathbf{N}(d_1)$ and the difference between $B(n \geq a|N, p)$ and $\mathbf{N}(d_2)$ vanish as N approaches infinity. Likewise, for put options, the convergence of P to $\mathbf{P}$ as N approaches infinity requires the vanishing of both the difference between $B(n < a|N, p')$ and $\mathbf{N}(-d_1)$ and the difference between $B(n < a|N, p)$ and $\mathbf{N}(-d_2)$. Given put-call parity, the convergence of C to $\mathbf{C}$ implies the convergence of P to $\mathbf{P}$, and vice versa.

To capture the volatility parameter σ in the BSM-OPM by the multiplicative up-down factors u and d for the underlying stock price over each of the N time steps in the CRR-BOPM, CRR have justified the use of

$$u = \frac{1}{d} = \exp\left(\sigma\sqrt{\frac{T}{N}}\right) \tag{19}$$

on statistical grounds.[6] As $ud = 1$, inequality (4) implies

$$0 < d < 1 < 1 + r_f < u, \tag{20}$$

where $r_f > 0$. For any given r, T, and σ, each of u, d, and $1 + r_f$ satisfying inequality (20) will converge to one — implying that the gap between u and d will narrow and that r_f will converge to zero — as N increases towards infinity.

[5] For any x, the expressions e^x and $\exp(x)$ are equivalent. In what follows, the choice between the two expressions is a trade-off between conciseness and notational clarity.

[6] See, for example, Feng and Kwan [2012] for a pedagogic explanation.

3.1 The Original Inequality for Determining the Number of Successful Binomial Trials

For N binomial trials, integer a represents the number of successful trials for determining each of $B(n \geq a|N,p)$, $B(n \geq a|N,p')$, $B(n < a|N,p)$, and $B(n < a|N,p')$ in equations (10) and (16) given the corresponding probabilities of success, which are p and p'. For any given r, σ, T, X, and S in the BSM-OPM, having good computational precision in all of a, p, and p' in the CRR-BOPM is crucial for illustrating the convergence of the two models. Accordingly, computational issues arising from the narrowing of the gap between u and d as N increases need to be addressed.

To illustrate, suppose that the term $X/(d^N S)$ in the numerator on the right hand side of inequality (9) is computed before its natural logarithm is taken. This is seldom problematic if N is not a large number. However, as $0 < d < 1$, if N is approaching $2,147,483,646$, which is the highest number of binomial trials that the Excel function BINOM.DIST can accommodate, the computed value of d^N may exceed the limit of $\approx 2.2 \times 10^{-308}$ for small numbers in Excel.

Given that Excel can store numbers with 15 digits of precision, as N increases, the gap between u and d narrows, the ratio u/d decreases, and the term $\ln(u/d)$ in the denominator on the right hand side of inequality (9) moves closer to $\ln(1) = 0$. This alone affects the computation of $\ln[x/(d^N S)]/\ln(u/d)$ for the determination of integer a. Furthermore, as N increases, r_f will converge to zero. The worsening of computational precision in $1 + r_f$ affects the precision of p and p' in equations (5) and (11), respectively.

3.2 An Equivalent Inequality

The above computational problem can easily be avoided by writing

$$\ln\left[X/(d^N S)\right] = \ln(X/S) - N\ln(d).$$

Here, writing $\ln(d^N)$ equivalently as $N\ln(d)$ is crucial, as doing so allows us to bypass the computation of the natural logarithm of any nearly zero number d^N if N is large. To obtain an equivalent expression of inequality (9), from which integer a is deduced, let us rewrite equation (19) equivalently as

$$N \ln u = \sigma\sqrt{NT} \tag{21}$$

and

$$N \ln d = -\sigma\sqrt{NT}. \tag{22}$$

Differencing the corresponding sides of equations (21) and (22) leads to

$$N \ln(u/d) = 2\sigma\sqrt{NT}.$$

Accordingly, inequality (9) can be expressed as

$$n > \frac{\ln(X/S) - N\ln(d)}{\ln(u/d)} = \frac{\sigma\sqrt{NT} + \ln(X/S)}{(2/N)\,\sigma\sqrt{NT}},$$

which is equivalent to

$$n > \frac{N}{2} + \frac{\sqrt{N}}{2}\left[\frac{\ln(X/S)}{\sigma\sqrt{T}}\right]. \tag{23}$$

An advantage of using inequality (23) to deduce integer a is that there are no computational problems even for a very narrow gap between u and d.

Inequality (12) must be satisfied for the CRR-BOPM to work as intended. A sufficient condition for it to hold is

$$0 \leq \frac{N}{2} + \frac{\sqrt{N}}{2}\left[\frac{\ln(X/S)}{\sigma\sqrt{T}}\right] \leq N.$$

This condition, which is equivalent to

$$\sqrt{N} \geq \pm\frac{\ln(X/S)}{\sigma\sqrt{T}},$$

leads to

$$N \geq \frac{1}{T}\left[\frac{\ln(X/S)}{\sigma}\right]^2. \tag{24}$$

Thus, for any given σ, T, X, and S, as long as N is large enough for inequality (24) to hold, the CRR-BOPM can still accommodate deep-in-the-money and deep-out-of-the money call and put options where S and X are very far apart.

3.3 A Measure to Counteract Deteriorations of Computational Precision in Determining the Probabilities of Success in Binomial Trials

To address the computational issue pertaining to p given equations (18) and (19), we can rewrite equation (5) as

$$p = \frac{\exp(rT/N) - \exp\left(-\sigma\sqrt{T/N}\right)}{\exp\left(\sigma\sqrt{T/N}\right) - \exp\left(-\sigma\sqrt{T/N}\right)} = \frac{\exp\left(rT/N + \sigma\sqrt{T/N}\right) - 1}{\exp\left(2\sigma\sqrt{T/N}\right) - 1}. \tag{25}$$

Notice that Taylor's series of $\exp(x)$ for any variable x is

$$\exp(x) = 1 + x + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} + \cdots$$

or, equivalently,

$$\exp(x) - 1 = x + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} + \cdots$$

As both $rT/N + \sigma\sqrt{T/N}$ and $2\sigma\sqrt{T/N}$ decrease with increasing N, we can approximate p very well by retaining only several terms in Taylor's expansions of $\exp\left(rT/N + \sigma\sqrt{T/N}\right)$ and $\exp\left(2\sigma\sqrt{T/N}\right)$ in equation (25) for any large integer N that approaches or equals $2,147,483,646$.

To address the computational issue pertaining to p', we first combine equations (5) and (11) as

$$p' = \frac{(1 + r_f - d)u}{(u - d)(1 + r_f)} = \frac{1 - d/(1 + r_f)}{1 - d/u},$$

which leads to

$$p' = \frac{\exp\left(-rT/N - \sigma\sqrt{T/N}\right) - 1}{\exp\left(-2\sigma\sqrt{T/N}\right) - 1}. \tag{26}$$

Both $-rT/N - \sigma\sqrt{T/N}$ and $-2\sigma\sqrt{T/N}$ increase with increasing N instead. We can also approximate p' with good precision, by using Taylor's expansions of $\exp\left(-rT/N - \sigma\sqrt{T/N}\right)$ and $\exp\left(-2\sigma\sqrt{T/N}\right)$ in the same manner.

4 An Excel-Based Illustration

The Excel file accompanying this note has two worksheets. The first worksheet, named **Fig1**, is for the illustration in Figure 1 of the this note. The second worksheet, named **More**, is for users to vary the input parameters to generate additional examples for illustrating the convergence of the CRR-BOPM to the BSM-OPM. In both worksheets, the input parameters for computing the corresponding call and put option prices include S, X, T, r, and σ for the two models, as well as different values of N for the CRR–BOPM. For the computations of p and p' in equations (25) and (26), respectively, Taylor's expansion of each exponential function contains a fixed number of terms of the variable involved.

	A	B	C	D	E	F	G
1	S		20	20	20	20	20
2	X		18	18	18	18	18
3	T		0.25	0.25	0.25	0.25	0.25
4	r		0.05	0.05	0.05	0.05	0.05
5	sig		0.40	0.40	0.40	0.40	0.40
6	N		1,000,000	10,000,000	100,000,000	1,000,000,000	2,147,483,646
7							
8	N min requirement		2.775210E-01	2.775210E-01	2.775210E-01	2.775210E-01	2.775210E-01
9							
10	u		1.000200E+00	1.000063E+00	1.000020E+00	1.000006E+00	1.000004E+00
11	d		9.998000E-01	9.999368E-01	9.999800E-01	9.999937E-01	9.999957E-01
12	rf		1.250000E-08	1.250000E-09	1.250000E-10	1.250000E-11	5.820677E-12
13							
14	a (computed)		4.997366E+05	4.999167E+06	4.999737E+07	4.999917E+08	1.073730E+09
15	a (integer)		499,737	4,999,168	49,997,366	499,991,671	1,073,729,617
16							
17	p		4.999813E-01	4.999941E-01	4.999981E-01	4.999994E-01	4.999996E-01
18	p prime		5.000812E-01	5.000257E-01	5.000081E-01	5.000026E-01	5.000018E-01
19							
20	comp-bin, S term		7.541162E-01	7.543954E-01	7.546513E-01	7.546641E-01	7.546732E-01
21	comp-bin, X term		6.870480E-01	6.873620E-01	6.876499E-01	6.876643E-01	6.876746E-01
22							
23	C (CRR)		2.869083E+00	2.869085E+00	2.869085E+00	2.869085E+00	2.869085E+00
24							
25	d1		6.893026E-01	6.893026E-01	6.893026E-01	6.893026E-01	6.893026E-01
26	d2		4.893026E-01	4.893026E-01	4.893026E-01	4.893026E-01	4.893026E-01
27							
28	norm, S term		7.546836E-01	7.546836E-01	7.546836E-01	7.546836E-01	7.546836E-01
29	norm, X term		6.876863E-01	6.876863E-01	6.876863E-01	6.876863E-01	6.876863E-01
30							
31	C (BSM)		2.869085E+00	2.869085E+00	2.869085E+00	2.869085E+00	2.869085E+00
32							
33	diff, S term for C		-5.674105E-04	-2.881868E-04	-3.227050E-05	-1.950092E-05	-1.039957E-05
34	diff, X term for C		-6.382875E-04	-3.242072E-04	-3.630692E-05	-2.194011E-05	-1.170040E-05
35	diff, C		-1.755231E-06	-4.988322E-07	-3.714137E-09	-2.086978E-09	-5.383143E-10
36							
37	cumu-bin, S term		2.458838E-01	2.456046E-01	2.453487E-01	2.453359E-01	2.453268E-01
38	cumu-bin, X term		3.129520E-01	3.126380E-01	3.123501E-01	3.123357E-01	3.123254E-01
39							
40	P (CRR)		6.454837E-01	6.454850E-01	6.454855E-01	6.454855E-01	6.454855E-01
41							
42	norm(-d1), S term		2.453164E-01	2.453164E-01	2.453164E-01	2.453164E-01	2.453164E-01
43	norm(-d2), X term		3.123137E-01	3.123137E-01	3.123137E-01	3.123137E-01	3.123137E-01
44							
45	P (BSM)		6.454855E-01	6.454855E-01	6.454855E-01	6.454855E-01	6.454855E-01
46							
47	diff: S term for P		5.674105E-04	2.881868E-04	3.227050E-05	1.950092E-05	1.039957E-05
48	diff: X term for P		6.382875E-04	3.242072E-04	3.630692E-05	2.194011E-05	1.170040E-05
49	diff: P		-1.755231E-06	-4.988322E-07	-3.714134E-09	-2.086977E-09	-5.383134E-10
50							
51							

Figure 1: An Excel-Based Illustration of the Convergence of the Cox-Ross-Rubinstein Binomial Option Pricing Model to the Black-Scholes-Merton Version for Call and Put Options

	A	B	C	D	E	F	G
52	rT/N+sig*sqrt(T/N)	1	2.000125E-04	6.324680E-05	2.000013E-05	6.324568E-06	4.315843E-06
53		2	2.000250E-08	2.000079E-09	2.000025E-10	2.000008E-11	9.313251E-12
54		3	1.333583E-12	4.216620E-14	1.333358E-15	4.216395E-17	1.339818E-17
55		4	6.668333E-17	6.667194E-19	6.666833E-21	6.666719E-23	1.445611E-23
56		5	2.667500E-21	8.433574E-24	2.666750E-26	8.432824E-29	1.247806E-29
57		6	8.892223E-26	8.889943E-29	8.889222E-32	8.888994E-35	8.975557E-36
58		7	2.540794E-30	8.032293E-34	2.539794E-37	8.031292E-41	5.533871E-42
59		8	6.352382E-35	6.350210E-39	6.349524E-43	6.349307E-47	2.985415E-48
60		9	1.411729E-39	4.462561E-44	1.411014E-48	4.461847E-53	1.431620E-54
61		10	2.823634E-44	2.822427E-49	2.822046E-54	2.821925E-59	6.178648E-61
62		sum	2.000325E-04	6.324880E-05	2.000033E-05	6.324588E-06	4.315852E-06
63							
64	2*sig*sqrt(T/N)	1	4.000000E-04	1.264911E-04	4.000000E-05	1.264911E-05	8.631675E-06
65		2	8.000000E-08	8.000000E-09	8.000000E-10	8.000000E-11	3.725290E-11
66		3	1.066667E-11	3.373096E-13	1.066667E-14	3.373096E-16	1.071850E-16
67		4	1.066667E-15	1.066667E-17	1.066667E-19	1.066667E-21	2.312965E-22
68		5	8.533333E-20	2.698477E-22	8.533333E-25	2.698477E-27	3.992952E-28
69		6	5.688889E-24	5.688889E-27	5.688889E-30	5.688889E-33	5.744310E-34
70		7	3.250794E-28	1.027991E-31	3.250794E-35	1.027991E-38	7.083288E-40
71		8	1.625397E-32	1.625397E-36	1.625397E-40	1.625397E-44	7.642579E-46
72		9	7.223986E-37	2.284425E-41	7.223986E-46	2.284425E-50	7.329806E-52
73		10	2.889594E-41	2.889594E-46	2.889594E-51	2.889594E-56	6.326850E-58
74		sum	4.000800E-04	1.264991E-04	4.000080E-05	1.264919E-05	8.631712E-06
75							
76	p		4.999813E-01	4.999941E-01	4.999981E-01	4.999994E-01	4.999996E-01
77							
78	-rT/N+sig*sqrt(T/N)	1	-2.000125E-04	-6.324680E-05	-2.000013E-05	-6.324568E-06	-4.315843E-06
79		2	2.000250E-08	2.000079E-09	2.000025E-10	2.000008E-11	9.313251E-12
80		3	-1.333583E-12	-4.216620E-14	-1.333358E-15	-4.216395E-17	-1.339818E-17
81		4	6.668333E-17	6.667194E-19	6.666833E-21	6.666719E-23	1.445611E-23
82		5	-2.667500E-21	-8.433574E-24	-2.666750E-26	-8.432824E-29	-1.247806E-29
83		6	8.892223E-26	8.889943E-29	8.889222E-32	8.888994E-35	8.975557E-36
84		7	-2.540794E-30	-8.032293E-34	-2.539794E-37	-8.031292E-41	-5.533871E-42
85		8	6.352382E-35	6.350210E-39	6.349524E-43	6.349307E-47	2.985415E-48
86		9	-1.411729E-39	-4.462561E-44	-1.411014E-48	-4.461847E-53	-1.431620E-54
87		10	2.823634E-44	2.822427E-49	2.822046E-54	2.821925E-59	6.178648E-61
88		sum	-1.999925E-04	-6.324480E-05	-1.999992E-05	-6.324548E-06	-4.315834E-06
89							
90	-2*sig*SQRT(T/N)	1	-4.000000E-04	-1.264911E-04	-4.000000E-05	-1.264911E-05	-8.631675E-06
91		2	8.000000E-08	8.000000E-09	8.000000E-10	8.000000E-11	3.725290E-11
92		3	-1.066667E-11	-3.373096E-13	-1.066667E-14	-3.373096E-16	-1.071850E-16
93		4	1.066667E-15	1.066667E-17	1.066667E-19	1.066667E-21	2.312965E-22
94		5	-8.533333E-20	-2.698477E-22	-8.533333E-25	-2.698477E-27	-3.992952E-28
95		6	5.688889E-24	5.688889E-27	5.688889E-30	5.688889E-33	5.744310E-34
96		7	-3.250794E-28	-1.027991E-31	-3.250794E-35	-1.027991E-38	-7.083288E-40
97		8	1.625397E-32	1.625397E-36	1.625397E-40	1.625397E-44	7.642579E-46
98		9	-7.223986E-37	-2.284425E-41	-7.223986E-46	-2.284425E-50	-7.329806E-52
99		10	2.889594E-41	2.889594E-46	2.889594E-51	2.889594E-56	6.326850E-58
100		sum	-3.999200E-04	-1.264831E-04	-3.999920E-05	-1.264903E-05	-8.631637E-06
101							
102	p prime		5.000812E-01	5.000257E-01	5.000081E-01	5.000026E-01	5.000018E-01

Figure 1: An Excel-Based Illustration of the Convergence of the Cox-Ross-Rubinstein Binomial Option Pricing Model to the Black-Scholes-Merton Version for Call and Put Options (Continued)

Figure 1 shows the computational details and comparisons of the corresponding option prices from the two models. The input parameters for the illustration are as follows: $S = \$20$, $X = \$18$, $T = 0.25$ years, $r = 5\% = 0.05$ each year, $\sigma = 40\% = 0.40$ each year, and $N = 10^6$, 10^7, 10^8, and 10^9, as well as $2,147,483,646$, which is the highest number of binomial trials that can be accommodated by the Excel function BINOM.DIST. These input parameters have been entered to the corresponding shaded (yellow) cells in C1:C5 and C6:G6. For computing p and p', 10 terms are retained in Taylor's expansion of each exponential function of the variable involved. All computed values except for integers are displayed in the scientific format.

All cell formulas for column C are listed in the Appendix of this note. The remaining four columns in Figure 1 differ from column C only in the choice of N, the number of time steps in the CRR-BOPM. To ensure that the remaining input parameters be common for all columns, the cell formula for D1, which is =$C1, has been pasted to D1:G5. All cell formulas in C8:C102 has been pasted to D8:G102.

The second worksheet differs from the first worksheet in three ways. (1) It considers 10 different values of N. (2) It retains 15 terms for Taylor's expansion of each exponential function. (3) More digits are displayed in each cell. In addition to user input for S, X, T, r, and σ in C1:C5, the 10 initial values of N, which are $10, 10^2, 10^3, \ldots, 10^9$, and $2,147,483,646$, also as user input, have been entered to the shaded (yellow) cells in C6:I6.

Given Excel's 15-digit precision, it is adequate to retain only 10 terms in Taylor's expansions of exponential functions where the variables involved typically have magnitudes of 10^{-3} or lower. However, the retention of five extra terms ensure that good computational precision will be maintained for cases where $N \leq 10^2$ in user input to the second worksheet. The computational approach in the two worksheets being the same, the descriptions below for the first worksheet in the Figure 1 pertain to the second worksheet as well, except for changes in some cell references.

In Figure 1, as all computations pertaining to a given N are in the same column, let us start with column C, where $N = 10^6$ is the lowest among the five values of N considered. The first step in the computations is to verify whether inequality (24) holds. The computed value of $[\sigma^{-1} \ln(X/S)]^2/T$, denoted as "N min requirement" and displayed in C8, is approximately 2.775×10^{-1}, which is far below $N = 10^6$ in C6. Thus, inequality (12) is assured. A further implication is that the remaining computations in column C would still be performed without problems should $N = 1$ be entered to C6 instead.

The computed values of u, d, and r_f based on equations (18) and (19) are displayed in C10:C12. To avoid deteriorations of computational precision as N increases towards its upper limit of $2,147,483,646$, these three values are not used for subsequent computations. For example, as the value of r_f in C12 for $N = 10^6$ is $1.25000001460762 \times 10^{-8}$ with 15-digit precision, the corresponding value of $1 + r_f$ is 1.00000001250000. The use of this truncated value of $1 + r_f$ will affect the precision of the computed values of p and p' slightly if equations (**??**) and (11), respectively, are used for the tasks involved. However, as increases in N will lead to lower values of r_f, deteriorations of computations precision in p and p' will be more severe.

The computed value of the term on the right hand side of inequality (23), denoted as "a (computed)," is displayed in C14. The smallest integer that is greater than the number in C14 is integer a; it is denoted as "a (integer)" and displayed in C15. The computed values of p and p' as displayed in C17:C18 are based on equations (25) and (26), respectively. The corresponding computations are provided in C52:C76 and C78:C102.

The block of cells C52:C62 is for computing the numerator $\exp\left(rT/N + \sigma\sqrt{T/N}\right) - 1$ in the expression of p in equation (25), by retaining 10 terms in Taylor's expansion of $rT/N + \sigma\sqrt{T/N}$, denoted as "p numer, Tay i," for $i = 1, 2, \ldots, 10$. The sum of the 10 terms,

$$\sum\nolimits_{i=1}^{10} \frac{\left(rT/N + \sigma\sqrt{T/N}\right)^i}{i!},$$

denoted as "p numer, sum," is displayed in C62. Likewise, C64:C74 is for computing the denominator $\exp\left(2\sigma\sqrt{T/N}\right) - 1$ in the expression of p in the same manner. The sum of the 10 corresponding terms, denoted as "p denom, Tay i," for $i = 1, 2, \ldots, 10$,

$$\sum\nolimits_{i=1}^{10} \frac{\left(2\sigma\sqrt{T/N}\right)^i}{i!},$$

denoted as "p denom, sum," is displayed in C74. The ratio of these two sums, which is the computed p as displayed in C76, is for C17.

The computations of the numerator $\exp\left(-rT/N - \sigma\sqrt{T/N}\right) - 1$ and the denominator $\exp\left(-2\sigma\sqrt{T/N}\right) - 1$ in the expression of p' in equation (26) are analogous. The only difference is the sign change of the variable involved in each case. The individual terms in the corresponding Taylor's expansions are denoted as "p pr numer, Tay i" and "p pr denom, Tay i," for $i = 1, 2, \ldots, 10$, and displayed in C78:C87 and C90:C99. The ratio of the corresponding sums, denoted as "p pr numer, sum" and "p pr denom, sum" and displayed in C88 and C100, is the computed p' in C102; it is for C18.

Now that a, p, and p' — all of which are shaded (in blue) — have been computed, the remaining computations are for various cells in C20:C49. To compute C by using equation (10) for C23, denoted as "C (CRR)," requires the use of the Excel function BINOM.DIST for cumulative binomial distributions to deduce the complementary binomial distributions in C20:C21. The latter distributions are denoted as "comp-bin, S term" for $B(n \geq a|N, p')$ and "comp-bin, X term" for $B(n \geq a|N, p)$.

To compute $\mathbf{C}$ in the BSM-OPM according to equation (1) requires that d_1 and d_2 in equations (2) and (3), respectively, to be computed first. The corresponding values of d_1, d_2, $\mathbf{N}(d_1)$, $\mathbf{N}(d_2)$, and $\mathbf{C}$, denoted as "d1," "d2," "norm, S term," "norm, X term," and "C (BSM)," respectively, are displayed in C25:C26, C28:C29, and C31. The differences between $B(n \geq a|N, p')$ and $\mathbf{N}(d_1)$, between $B(n \geq a|N, p)$ and $\mathbf{N}(d_2)$, and between C and $\mathbf{C}$, denoted as "diff, S term for C," "diff, X term for C," and "diff, C," respectively, are displayed in C33:C35, with C35 shaded (in orange).

Computations and comparisons of put option prices $\mathbf{P}$ and P in equations (14) and (16), respectively, in C37:C49 are analogous to what has been described above. The computed values of $B(n < a|N, p')$, $B(n < a|N, p)$, and P, denoted as "cumu-bin, S term," "cumu-bin, X term," and "P (CRR)," respectively, are displayed in C37:38 and C40. The computed values of $\mathbf{N}(-d_1)$, $\mathbf{N}(-d_2)$, and $\mathbf{P}$, denoted as "norm(-d1), S term," "norm(-d2), X term," and "P (BSM)," respectively, are displayed in C42:C43 and C45. The differences between $B(n < a|N, p')$ and $\mathbf{N}(-d_1)$, between $B(n < a|N, p)$ and $\mathbf{N}(-d_2)$, and between P and $\mathbf{P}$, denoted as "diff, S term for P," "diff, X term for P," and "diff, P," respectively, are displayed in C47:C49, with C49 also shaded (in orange).

As the computations in columns D, E, F, and G of the worksheet are the same as those in column C, a comparison of the computed results in the five columns reveals how they vary with increasing N. It is clear that, for the set of input parameters in Figure 1, as N increases, the computed option prices from the CRR-BOPM are getting closer to the corresponding option prices from the BSM-OPM. To assess the robustness of the results will require various sets of input parameters to be attempted. The second worksheet, named **More**, is suitable for this purpose.

5 Concluding Remarks

The Cox-Ross-Rubinstein binomial option pricing model (CRR-BOPM) is highly valuable for teaching option pricing, as it has captured the economic insights of the Black-Scholes-Merton option pricing model (BSM-OPM) without the encumbrance of any sophisticated mathematical tools. Finance students who are familiar with the binomial theorem in algebra — which can be explained intuitively in terms of the growth pattern of Pascal's triangle — are expected to follow the derivation of the CRR-BOPM very well. The algebraic relationship that connects the input parameters for the two models, as provided in the CRR-BOPM, is simple enough that the convergence of the two models can be examined by using an easily constructed Excel worksheet.

From a statistical standpoint, the more time steps between the valuation date and the expiry date of an option in the CRR-BOPM, as captured by increases of the integer N, which provides the number of binomial trials in the Excel function BINOM.DIST, the closer is the model in approximating the BSM-OPM. This statistical property justifies the use of large values of N to illustrate the convergence of the two models. However, increases in N can potentially cause deteriorations of computational precision.

The input parameters for computing call and put option prices according to the BSM-OPM include S, X, T, r, and σ. They are for the underlying stock price, the exercise price, the proportion of a year before the option expires, the continuously compounded annual risk-free interest rate, and the standard deviation of annual stock returns, respectively. Computations of call and put option prices in equations (1) and (14), by using the Excel function NORM.DIST, are straightforward.

The input parameters in the CRR-BOPM include u, d, r_f, p, and p' — for the multiplicative up-down factors in each time step, the risk-free interest rate in each time step, and the two probabilities of success in binomial trials, respectively — are connected to S, X, T, r, σ, and N. This note shows that the computations of call and put option prices in equations (10) and (16) can be entirely based on the latter set of six input parameters. The approach in this note, which uses Taylor's expansions of some exponential functions to compute p and p', is effective in preventing potential deteriorations of computational precision due to the narrowing of the gap between u and d as N increases.

There is flexibility in the delivery of the CRR-BOPM in finance courses where the topic of options is covered. For classes where students are knowledgeable of the binomial theorem in algebra, the coverage of the CRR-BOPM can include a self-contained model derivation. If such an approach is adopted, as the model derivation for a European call option on a stock that pays no dividends is similar to that for the corresponding put option, either derivation can be left as an exercise for students.

Another approach is to capture the essence of the derivation of the CRR-BOPM by considering $N = 1$ and $N = 2$ only. Such an approach still allows the idea of risk-free hedges in the model derivation to be illustrated. Alternatively, the model derivation can be omitted entirely. The corresponding coverage of the CRR-BOPM will be just like the coverage of the BSM-OPM in introductory finance courses, where the attention is on the derived option pricing formulas. If such an approach is adopted, students can still gain valuable experience from computational exercises. The Excel file accompanying this note can be used to generate exercises with various sets of input parameters and the corresponding answer keys.

Appendix: Representative Cell Formulas in an Excel Worksheet

The following cell formulas are for column C of worksheet **Fig1**:

- C8, =((LN(C2/C1)/C5)^2)/C3
- C10, =EXP(C5*SQRT(C3/C6))
- C11, =1/C10
- C12, =EXP(C4*C3/C6)-1
- C14, =(C6+SQRT(C6/C3)*LN(C2/C1)/C5)/2
- C15, =INT(C14+1)
- C17, =C76
- C18, =C102
- C20, =1-BINOM.DIST(C15,C6,C18,TRUE)

- C21, =1-BINOM.DIST(C15,C6,C17,TRUE)
- C23, =C1*C20-C2*EXP(-C4*C3)*C21
- C25, =(LN(C1/C2)+C4*C3)/C5/SQRT(C3)+C5*SQRT(C3)/2
- C26, =C25-C5*SQRT(C3)
- C28, =NORM.DIST(C25,0,1,TRUE)
- C29, =NORM.DIST(C26,0,1,TRUE)
- C31, =C1*C28-C2*EXP(-C4*C3)*C29
- C33, =C20-C28
- C34, =C21-C29
- C35, =C23-C31
- C37, =BINOM.DIST(C15,C6,C18,TRUE)
- C38, =BINOM.DIST(C15,C6,C17,TRUE)
- C40, =C2*EXP(-C4*C3)*C38-C1*C37
- C42, =NORM.DIST(-C25,0,1,TRUE)
- C43, =NORM.DIST(-C26,0,1,TRUE)
- C45, =C2*EXP(-C4*C3)*C43-C1*C42
- C47, =C37-C42
- C48, =C38-C43
- C49, =C40-C45
- C52, =C4*C3/C6+C5*SQRT(C3/C6)
- C53, =(C$52^$B53)/FACT($B53) — pasted to C53:C61

- C62, =SUM(C52:C61)
- C64, =2*C5*SQRT(C3/C6)
- C65, =(C$64^$B65)/FACT($B65) — pasted to C65:C73
- C74, =SUM(C64:C73)
- C76, =C62/C74
- C78, =-C4*C3/C6-C5*SQRT(C3/C6)
- C79, =(C$78^$B79)/FACT($B79) — pasted to C79:C87
- C88, =SUM(C78:C87)
- C90, =-2*C5*SQRT(C3/C6)
- C91, =(C$90^$B91)/FACT($B91) — pasted to C91:C99
- C100, =SUM(C90:C99)
- C102, =C88/C100

Acknowledgements

The author wishes to thank the three anonymous reviewers for valuable comments and suggestions.

References

Berk, J., and DeMarzo, P., [2020], *Corporate Finance*, *Fifth Edition*, Pearson Education, Inc., Hoboken, NJ, U.S.A.

Berk, J., DeMarzo, P., and Stangeland, D., [2022], *Corporate Finance*, *Fifth Canadian Edition*, Pearson Canada Inc., North York, ON, Canada.

Black, F., and Scholes, M., [1973], "The Pricing of Options and Corporate Liabilities," *Journal of Political Economy*, 81(3), 637-654.

Chance, D.M., [2008], "A Synthesis of Binomial Option Pricing Models for Lognormally Distributed Assets," *Journal of Applied Finance*, Spring/Summer, 38-56.

Copeland, T.E., Weston, J.F., and Shastri, K., [2005], *Financial Theory and Corporate Policy, Fourth Edition*, Pearson Addition Wesley, U.S.A.

Cox, J.C., Ross, S.A., and Rubinstein, M., [1979], "Option Pricing: A Simplified Approach," *Journal of Financial Economics*, 7, 229-263.

Feng, Y., and Kwan, C.C.Y., [2012], "Connecting Binomial and Black-Scholes Option Pricing Models: A Spreadsheet-Based Illustration," *Spreadsheets in Education*, 5(3), Article 2.

Hull, J.C., [2018], *Options, Futures, and Other Derivatives, Tenth Edition*, Pearson, New York, NY, U.S.A.

Merton, R.C., [1973], "Theory of Rational Option Pricing," *Rand Journal of Economics and Management Science*, 4(1), 141-183.

Ross, S.A., Westerfield, R.W., Jaffe, J., and Driss, H., [2022], *Corporate Finance, Ninth Canadian Edition*, McGraw-Hill Ryerson, Toronto, ON, Canada.

Ross, S.A., Westerfield, R.W., Jaffe, J., and Jordan, B.D. [2022], *Corporate Finance, Thirteenth Edition*, McGraw-Hill/Irwin, New York, NY, U.S.A.